FT62F21X

Application note

目录

FT62F21X IR Send 应用

1 IR 介绍

一个通用的红外遥控系统由发射和接收两大部分组成，如图 1 所示：



图 1

发射部分主要包括键盘矩阵、编码调制、红外发射管； 接收部分包括光、电信号的转换以及放大、解调、解码电路。

举例来说，通常我们家电遥控器信号的发射，就是将相应按键所对应的控制指令和系统码(由 0 和 1 组成的序列)，调制在 32~56kHz 范围内的载波上（目的为：抗干扰及低功率），然后经放大（接三极管）、驱动红外发射管（透明的头）将信号发射出去。

2 IR Send 相关寄存器的设置

本例使用两个定时器，一个是产生38KHz载波频率，另一个定时器是做时基，定时时长是560uS，红外信号的高低电平是560uS的整数倍。

定时器 0 为 8 位，可配置为计数器或定时器使用，当作为外部事件（T0CKI）计数器时，可以配置为上升沿或者下降沿计数。作为定时器时，其计数时钟可由 T0CKSRC 控制，选择时钟源来进行计数。有一个与 WDT 共用的 8 位预分频器，PSA 为 0 时该预分频器分配给定时器 0 使用。

例如在系统时钟和 4T 模式下，定时时长计算公式如下：

$$\text{定时时长}=\frac{1}{\text{系统时钟频率}}*4*\text{预分频值}*255$$

定时器 2 为 16 位，其时钟源由 T2CKSRC 控制，可以作为计数器和定时器使用，当 TMR2 值等于 PR2 时会产生中断，Timer2 具有预分频器和后分频器，预分频比为 1：1、1：4 和 1：16，后分频比为 1：1~1：16。

在系统时钟和 4T 模式下，定时时长计算公式如下：

$$\text{定时时长}=\frac{1}{\text{系统时钟频率}}*4*\text{预分频值}*\text{后分频值}*PR2$$

本讲解以IC FT62F211 SOP8为示范，每5秒钟会发出一次信号，信号的码为IRData[4] = {0x00,0xff,0x40,0xBf}

本程序IR接收与LED所对应的IO引脚：

```
#define    SendIO          RA4
```

3 应用范例

```
//=============================================================
;*    文件名：   ASM_FT62F21X_IR_Send.ASM
;*    功能：     FT62F21X  红外发送  功能演示
;*    IC:        FT62F211        SOP8
;*    内部：     16M/4T
;*    说明：     当按键按下后，发送自己所想要发送的红外码，按键不松开时发送连发码，
;*               此程序所发红外码为 ox55，oxaa，0x01，0xfe
;*
;*
;*                      FT62F211    SOP8
;*                     -------------------
;* irdata---------|1(PA4)     (PA3)10|-----------key
;* NC-------------|2(TKCAP) (PA0)9|-----------NC
;* VDD------------|3(VDD)    (PA1)8|-----------NC
;* GND------------|4(VSS)     (PA2)7|-----------NC
;*                      -------------------
//=============================================================
#INCLUDE    <FT62F21X.INC>;
;=============================================================
;RAM DEFINE
;=============================================================
      TEMP                EQU         0X40
      TEMP1               EQU         0X41
      TEMP2               EQU         0X42
      IRDATTEMP           EQU         0X43
      READPIN             EQU         0X44
      buff                EQU         0X45
      #DEFINE             f_2ms       buff,0
      #DEFINE             f_key       buff,1
      key_short_count     EQU         0X46
      count               EQU         0x47
      countbyte           EQU         0x48

      W_TMP               EQU         0X70
      S_TMP               EQU         0X71
;=============================================================
;CONSTANT DEFINE
;=============================================================
      INTCON_DEF      EQU      B'00000000'    ;禁止所有中断
```

```
    OSCCON_DEF    EQU    B'01110000'   ;16MHz

    WPUA_DEF      EQU    B'00001000'   ;弱上拉的开关，0-关，1-开

    TRISA_DEF     EQU    B'00001000'   ;输入输出设置，0-输出，1-输入

    PSRCA_DEF     EQU    B'00001111'   ;源电流设置最大

    PSINKA_DEF    EQU    B'00000011'   ;灌电流设置最大

    OPTION_DEF    EQU    B'00001000'   ;Bit3=1 WDT MODE,PS=000=1:1 WDT RATE
                                       ;Bit7(PAPU)=0  由 WPUA 决定是否上拉
;=============================================================
;USER DEFINE
;=============================================================
#DEFINE   ir_data     PORTA,4
#DEFINE   key         PORTA,3
;=============================================================
;PROGRAM START
;=============================================================
    ORG         0x0000          ; 单片机复位向量入口
    LJUMP       RESTART         ; 跳转到主程序入口
    ORG         0x0004          ; 中断复位向量入口
    LJUMP       INT_PROGRAM
;=============================================================
;中断处理程序
;=============================================================
INT_PROGRAM:
    STR         W_TMP           ; 保存 W 寄存器
    SWAPR       STATUS,W        ; 保存 STATUS 寄存器
    STR         S_TMP

    BANKSEL     INTCON
    BTSC        INTCON,T0IF
    LJUMP       TM0Interrupt
    BTSC        INTCON,PAIF
    LJUMP       GPIOInterrupt
    LJUMP       INT_RET
TM0Interrupt:
    BCR         INTCON,T0IF
    BSR         f_2ms
    LJUMP       INT_RET
GPIOInterrupt:
```

```
    BANKSEL   PORTA
    LDR       PORTA,W
    BANKSEL   READPIN
    STR       READPIN
    BANKSEL   INTCON
    BCR       INTCON,PAIF
    BCR       INTCON,PAIE
    BANKSEL   IOCA
    BCR       IOCA,IOCA3
    LJUMP     INT_RET
INT_RET:
    SWAPR     S_TMP,0
    STR       STATUS          ; 恢复 STATUS 寄存器
    SWAPR     W_TMP,1
    SWAPR     W_TMP,0         ; 恢复 W 寄存器
    RETI                      ; 中断返回
;=========================================================
;SYSTEM START
;=========================================================
RESTART:
    BANKSEL   PORTA
    LCALL     INITIAL
    LCALL     TIMER0_INITIAL
;=========================================================
;主程序
;=========================================================
MAIN:
    CLRWDT
    BTSS      f_2ms
    LJUMP     MAIN
    BCR       f_2ms                ;主程序 2ms 扫描一次
    LCALL     scanky_key
    ;LCALL    sleep_mode
    LJUMP     MAIN
;=========================================================
;系统初始化
;=========================================================
INITIAL:
    BANKSEL   OSCCON
    LDWI      OSCCON_DEF
    STR       OSCCON

    BANKSEL   INTCON
    LDWI      INTCON_DEF
```

```
    STR     INTCON

    BANKSEL PORTA
    LDWI    0X20
    STR     PORTA

    BANKSEL TRISA
    LDWI    TRISA_DEF
    STR     TRISA

    BANKSEL WPUA
    LDWI    WPUA_DEF
    STR     WPUA

    BANKSEL PSRCA
    LDWI    PSRCA_DEF
    STR     PSRCA

    BANKSEL PSINKA
    LDWI    PSINKA_DEF
    STR     PSINKA

    BANKSEL OPTION
    LDWI    OPTION_DEF
    STR     OPTION
;***************Clear  SRAM***********************************
    BCR     STATUS,PAGE
    LDWI    0X40
    STR     FSR
CLEAR_RAM_BANK0_LOOP:
    CLRR    INDF
    INCR    FSR,F
    LDWI    80H
    XORWR   FSR,W
    BTSS    STATUS,Z
    LJUMP   CLEAR_RAM_BANK0_LOOP
    RET
;=========================================================
;PA3_Level_Change_INITIAL
;=========================================================
PA3_Level_Change_INITIAL:
    BANKSEL TRISA
    BSR     TRISA,3                 ;端口设置为输入
```

```
    BANKSEL   PORTA
    LDR       PORTA,W
    BANKSEL   READPIN
    STR       READPIN

    BANKSEL   INTCON
    BCR       INTCON,PAIF          ;中断标志清零
    BANKSEL   IOCA
    BSR       IOCA,IOCA3
    BANKSEL   INTCON
    BSR       INTCON,PAIE          ;中断使能
    RET
;=======================================================
;TIMER0_INITIAL
;设置 TMR0 定时时长 2.040ms=(1/16000000)*4*32*255
;=======================================================
TIMER0_INITIAL:
    BANKSEL   OPTION
    LDWI      0X04
    STR       OPTION
    BANKSEL   TMR0
    LDWI      0
    STR       TMR0
    BANKSEL   INTCON
    BSR       INTCON,T0IE
    BSR       INTCON,GIE
    RET
;=======================================================
;DELAY_8US 16MHZ/4T   (实测 7.5US)
;=======================================================
DELAY_8US:
    LDWI      0x06
    STR       TEMP
DELAY_8USLOOP:
    CLRWDT
    DECRSZ    TEMP,F
    LJUMP     DELAY_8USLOOP
    RET
;=======================================================================
;DELAY_18US 16MHZ/4T    (实测 17.5US)
;=======================================================================
DELAY_18US:
    LDWI      0x10
    STR       TEMP
```

```
DELAY_18USLOOP:
    CLRWDT
    DECRSZ    TEMP,F
    LJUMP     DELAY_18USLOOP
    RET
;=========================================================
; IR_start
; 红外发送的引导码 9ms 发送 4.5ms 停止
;=========================================================
IR_start:
    CLRR   count
IR_start_working:
    LDWI   0xa2       ;9000=173*52 us
    SUBWR  count,0
    BTSC   STATUS,0
    LJUMP  IR_start_no_work
    INCR   count,1
    BSR    ir_data
    LCALL  DELAY_8US
    BCR    ir_data
    LCALL  DELAY_18US
    BSR    ir_data
    LCALL  DELAY_8US
    BCR    ir_data
    LCALL  DELAY_18US
    LJUMP  IR_start_working
IR_start_no_work:
    CLRR   count
IR_start_loop:
    LDWI   0x50       ;4500=86*52 us
    SUBWR  count,0
    BTSC   STATUS,0
    RET
    INCR   count,1
    BCR    ir_data
    LCALL  DELAY_8US
    BCR    ir_data
    LCALL  DELAY_18US
    BCR    ir_data
    LCALL  DELAY_8US
    BCR    ir_data
    LCALL  DELAY_18US
    LJUMP  IR_start_loop
;=========================================================
```

```
; IR_stop
;红外发送的连发码  9ms  发送 2.25ms 停止
;=============================================================
IR_stop:
    CLRR    count
IR_stop_working:
    LDWI    0xa2        ;9000=173*52 us
    SUBWR   count,0
    BTSC    STATUS,0
    LJUMP   IR_stop_no_work
    INCR    count,1
    BSR     ir_data
    LCALL   DELAY_8US
    BCR     ir_data
    LCALL   DELAY_18US
    BSR     ir_data
    LCALL   DELAY_8US
    BCR     ir_data
    LCALL   DELAY_18US
    LJUMP   IR_stop_working
IR_stop_no_work:
    CLRR    count
IR_stop_loop:
    LDWI    0x28        ;2250=43*52 us
    SUBWR   count,0
    BTSC    STATUS,0
    RET
    INCR    count,1
    BCR     ir_data
    LCALL   DELAY_8US
    BCR     ir_data
    LCALL   DELAY_18US
    BCR     ir_data
    LCALL   DELAY_8US
    BCR     ir_data
    LCALL   DELAY_18US
    LJUMP   IR_stop_loop
;=============================================================
; IR_Send_Byte
;红外发送一个字节
;=============================================================
IR_Send_Byte:
    CLRR    countbyte
IR_Send_Byte_loop:
```

```
    LDWI    0x08
    SUBWR   countbyte,0
    BTSC    STATUS,0
    RET
    INCR    countbyte,1
    BTSS    IRDATTEMP,0
    LJUMP   $+3
    LCALL   Send_IRdata1
    LJUMP   $+2
    LCALL   Send_IRdata0

    BCR     STATUS,0
    RRR     IRDATTEMP,1
    LJUMP   IR_Send_Byte_loop
;=============================================================
; Send_IRdata0
;红外发送数据 0
;=============================================================
Send_IRdata0:
    CLRR    count
Send_IRdata0_working:
    LDWI    0x11        ;560=22*26 us
    SUBWR   count,0
    BTSC    STATUS,0
    LJUMP   Send_IRdata0_no_work
    INCR    count,1
    BSR     ir_data
    LCALL   DELAY_8US
    BCR     ir_data
    LCALL   DELAY_18US
    LJUMP   Send_IRdata0_working
Send_IRdata0_no_work:
    CLRR    count
Send_IRdata0_loop:
    LDWI    0x11        ;560=22*26 us
    SUBWR   count,0
    BTSC    STATUS,0
    RET
    INCR    count,1
    BCR     ir_data
    LCALL   DELAY_8US
    BCR     ir_data
    LCALL   DELAY_18US
    LJUMP   Send_IRdata0_loop
```

```
;=======================================================
; Send_IRdata1
;红外发送数据 1
;=======================================================
Send_IRdata1:
    CLRR    count
Send_IRdata1_working:
    LDWI    0x11        ;560=22*26 us
    SUBWR   count,0
    BTSC    STATUS,0
    LJUMP   Send_IRdata1_no_work
    INCR    count,1
    BSR     ir_data
    LCALL   DELAY_8US
    BCR     ir_data
    LCALL   DELAY_18US
    LJUMP   Send_IRdata1_working
Send_IRdata1_no_work:
    CLRR    count
Send_IRdata1_loop:
    LDWI    0x37        ;1685=65*26 us
    SUBWR   count,0
    BTSC    STATUS,0
    RET
    INCR    count,1
    BCR     ir_data
    LCALL   DELAY_8US
    BCR     ir_data
    LCALL   DELAY_18US
    LJUMP   Send_IRdata1_loop
;=============================================================================
;scanky_key
;当按键按下时候发送所需要发送的红外数字，长按发送连发码
;=============================================================================
 scanky_key:
    BTSS    key
    LJUMP   scankey_key_down
    LJUMP   scankey_key_up
scankey_key_down:
    BTSC    f_key
    LJUMP   scankey_key_long
    INCR    key_short_count,1
    LDWI    0x0a
    SUBWR   key_short_count,0
```

```
    BTSS    STATUS,0
    RET
    CLRR    key_short_count

    BSR     f_key

    LCALL   IR_start
    LDWI    0x55
    STR     IRDATTEMP
    LCALL   IR_Send_Byte
    LDWI    0xaa
    STR     IRDATTEMP
    LCALL   IR_Send_Byte
    LDWI    0x01
    STR     IRDATTEMP
    LCALL   IR_Send_Byte
    LDWI    0xfe
    STR     IRDATTEMP
    LCALL   IR_Send_Byte
    LCALL   Send_IRdata0
    RET
scankey_key_long:
    LCALL   IR_stop
    RET
scankey_key_up:
    CLRR    key_short_count
    BCR     f_key
    RET
;=============================================================
; sleep_mode
;无操作进入睡眠
;=============================================================
sleep_mode:
    BTSS    key
    RET
    BTSC    f_key
    RET

    LCALL   PA3_Level_Change_INITIAL
    SLEEP
    NOP
    RET
;=============================================================
    END                                 ; 汇编程序结束
```

Fremont Micro Devices (SZ) Limited

#5-8, 10/F, Changhong Building, Ke-Ji Nan 12 Road, Nanshan District, Shenzhen, Guangdong 518057
Tel: (86 755) 86117811
Fax: (86 755) 86117810

Fremont Micro Devices (Hong Kong) Limited

#16, 16/F, Blk B, Veristrong Industrial Centre, 34-36 Au Pui Wan Street, Fotan, Shatin, Hong Kong
Tel: (852) 27811186
Fax: (852) 27811144

Fremont Micro Devices (USA), Inc.

42982 Osgood Road Fremont, CA 94539
Tel: (1-510) 668-1321
Fax: (1-510) 226-9918

Web Site: http://www.fremontmicro.com/