

FT62F21X

Application note

目录

| | | |
|---|----------------------------|---|
| 1 | PA2/INT 外部中断相关寄存器的设置 | 3 |
| 2 | 应用范例 | 5 |

FT62F21x INT 外部中断的应用

1 PA2/INT 外部中断相关寄存器的设置

INT 引脚上的外部中断是边沿触发的；当 OPTION 寄存器的 INTEDG 位被置 1 时在上升沿触发，而当 INTEDG 位被清零时在下降沿触发。当 INT 引脚上出现有效边沿时，INTCON 寄存器的 INTF 位置 1。可以通过将 INTCON 寄存器的 INTE 控制位清零来禁止该中断。在重新允许该中断前，必须在中断服务程序中先用软件将 INTF 位清零。如果 INTE 位在进入休眠状态前被置 1，则 INT 中断能将处理器从休眠状态唤醒。

相关寄存器的各个位定义如下：

1) PORTA 寄存器

| Bit | 7 | 6 | 5 | 4 | 3 | 2 | 1 | 0 |
|-------|---|---|-----|-----|-----|-----|-----|-----|
| Name | - | - | PA5 | PA4 | PA3 | PA2 | PA1 | PA0 |
| Reset | - | - | x | x | x | x | x | x |

Bit7~Bit6: 保留位，读 0

Bit5: PORTA5 数据

Bit4: PORTA4 数据

Bit3: PORTA3 数据

Bit2: PORTA2 数据

Bit1: PORTA1 数据

Bit0: PORTA0 数据

2) TRISA 寄存器

| Bit | 7 | 6 | 5 | 4 | 3 | 2 | 1 | 0 |
|-------|---|---|----------|----------|----------|----------|----------|----------|
| Name | - | - | TRISA[5] | TRISA[4] | TRISA[3] | TRISA[2] | TRISA[1] | TRISA[0] |
| Reset | - | - | 1 | 1 | 1 | 1 | 1 | 1 |

Bit7~Bit6: N/A, 读 0

Bit5~Bit0: PORTA<5:0>输入/输出状态控制寄存器

1: 端口为输入状态

0: 端口为输出状态

3) WPUA 寄存器

| Bit | 7 | 6 | 5 | 4 | 3 | 2 | 1 | 0 |
|-------|---|---|-------|-------|-------|-------|-------|-------|
| Name | - | - | WPUA5 | WPUA4 | WPUA3 | WPUA2 | WPUA1 | WPUA0 |
| Reset | - | - | 1 | 1 | 1 | 1 | 1 | 1 |

Bit7~Bit6: N/A, 读 0

Bit5 ~Bit0: Port A 弱上拉使能

1: 使能PORTA 端口弱上拉

0: 关闭PORTA 端口弱上拉

4) OPTION 寄存器

| Bit | 7 | 6 | 5 | 4 | 3 | 2 | 1 | 0 |
|-------|-------|--------|------|------|-----|-----|-----|-----|
| Name | /PAPU | INTEDG | T0CS | T0SE | PSA | PS2 | PS1 | PS0 |
| Reset | 1 | 1 | 1 | 1 | 1 | 1 | 1 | 1 |

Bit7: PORTA 口上拉使能位

- 1: 禁止上拉
- 0: 由 WPUA 决定是否上拉

Bit6: 触发中断边沿选择位

- 1: PA2/INT 上升沿触发中断
- 0: PA2/INT 下降沿触发中断

Bit5: Timer0 时钟选择位

- 1: PA2/T0CKI管脚输入时钟
- 0: 时钟由T0CKSRC寄存器位决定

Bit4: Timer0 时钟边沿选择位

- 1: PA2/T0CKI管脚由高到底变化时计数增加
- 0: PA2/T0CKI管脚由低到高变化时计数增加

Bit3: 预分频分配位

- 1: 预分频器分配给WDT
- 0: 预分频器分配给Timer0

Bit2~Bit0: 预分频大小设置位

| Bit2: Bit0 | Timer0 Rate | WDT Rate |
|------------|-------------|----------|
| 000 | 1 : 2 | 1 : 1 |
| 001 | 1 : 4 | 1 : 2 |
| 010 | 1 : 8 | 1 : 4 |
| 011 | 1 : 16 | 1 : 8 |
| 100 | 1 : 32 | 1 : 16 |
| 101 | 1 : 64 | 1 : 32 |
| 110 | 1 : 128 | 1 : 64 |
| 111 | 1 : 256 | 1 : 128 |

5) INTCON 寄存器

| Bit | 7 | 6 | 5 | 4 | 3 | 2 | 1 | 0 |
|-------|-----|------|------|------|------|------|------|------|
| Name | GIE | PEIE | TOIE | INTE | PAIE | TOIF | INTF | PAIF |
| Reset | 0 | 0 | 0 | 0 | 0 | 0 | 0 | 0 |

Bit7: 全局中断使能位

- 1: 使能全局中断
- 0: 关闭全局中断

Bit6: 外设中断使能位

- 1: 使能外设中断
- 0: 关闭外设中断

Bit5: Timer0 定时器溢出中断使能位

- 1: 使能Timer0中断
- 0: 关闭Timer0中断

Bit4: 外部中断使能位

- 1: 使能PA2/INT外部中断
- 0: 关闭PA2/INT外部中断

Bit3: PORTA端口状态变化中断使能位

- 1: 使能PORTA端口状态变化中断使能

- 0: 关闭PORTA端口状态变化中断使能
- Bit2: Timer0定时器溢出中断标志位
- 1: Timer0寄存器溢出（必须软件清零）
- 0: Timer0寄存器无溢出
- Bit1: 外部中断标志位
- 1: PA2/INT外部中断发生（必须软件清零）
- 0: PA2/INT外部中断无发生
- Bit0: PORTA端口状态改变中断标志位
- 1: PORTA<5:0>至少有一个端口状态发生了改变（必须软件清零）
- 0: PORTA<5:0>没有一个端口发生状态改变

2 应用范例

```
//=====
;* 文件名:  ASM_FT62F21X_INT.ASM
;* 功能:    FT62F21X_INT 功能演示
;* IC:      FT62F211      SOP8
;* 内部:    16M/4T
;* 说明:    程序中 DemoPortOut(PA4)输出 100 帧 50HZ 的占空比为 50%的方波后,
;*          MCU 进入睡眠, 等待外部中断的发生;当外部中断触发后, 重复以上流程;
;*
;*
;*
;*          FT62F211  SOP8
;*          -----
;* DemoPortOut---|1(PA4)   (PA3)10|-----NC
;* NC-----|2(TKCAP)  (PA0)9|-----NC
;* VDD-----|3(VDD)   (PA1)8|-----NC
;* GND-----|4(VSS)   (PA2)7|-----INT
;*          -----
//=====
#include <FT62F21X.INC>;
;=====
;RAM DEFINE
;=====
TEMP      EQU      0X40
TEMP1     EQU      0X41
TEMP2     EQU      0X42
FCount    EQU      0X43
W_TMP     EQU      0X70
S_TMP     EQU      0X71
;=====
;CONSTANT DEFINE
;=====
INTCON_DEF EQU      B'00000000' ;禁止所有中断
```

```

OSCCON_DEF    EQU    B'01110000'    ;16MHz

WPUA_DEF      EQU    B'00000100'    ;弱上拉的开关, 0-关, 1-开

TRISA_DEF     EQU    B'00000100'    ;输入输出设置, 0-输出, 1-输入

PSRCA_DEF     EQU    B'00001111'    ;源电流设置最大

PSINKA_DEF    EQU    B'00000011'    ;灌电流设置最大

OPTION_DEF    EQU    B'01001000'    ;Bit3=1 WDT MODE,PS=000=1:1 WDT RATE
                                ;Bit6(INTEDG)=1 PA2/INT 上升沿产生中断
                                ;Bit7(PAPU)=0 由 WPUA 决定是否上拉
;=====
;USER DEFINE
;=====
#define DemoPortOut    PORTA,4
;=====
;PROGRAM START
;=====
    ORG        0x0000            ; 单片机复位向量入口
    LJUMP      RESTART          ; 跳转到主程序入口
    ORG        0x0004            ; 中断复位向量入口
    LJUMP      INT_PROGRAM
;=====
;中断处理程序
;=====
INT_PROGRAM:
    STR        W_TMP            ; 保存 W 寄存器
    SWAPR      STATUS,W         ; 保存 STATUS 寄存器
    STR        S_TMP

    BANKSEL    INTCON
    BCR        INTCON,INTF
    BCR        INTCON,INTE

INT_RET:
    SWAPR      S_TMP,0
    STR        STATUS          ; 恢复 STATUS 寄存器
    SWAPR      W_TMP,1
    SWAPR      W_TMP,0         ; 恢复 W 寄存器
    RETI                    ; 中断返回
;=====

```

```
;SYSTEM START
```

```
;
```

```
RESTART:
```

```
    BANKSEL    PORTA
    LCALL      INITIAL
```

```
;
```

```
;主程序
```

```
;
```

```
MAIN:
```

```
    BANKSEL    FCount
    LDWI        0X00
    STR         FCount
```

```
MAIN_LOOP:
```

```
    BSR        DemoPortOut
    LCALL      Delay_10MS
    BCR        DemoPortOut
    LCALL      Delay_10MS
    INCR       FCount,1
    LDWI       0X64
    SUBWR      FCount,W
    BTSS       STATUS,Z
    LJUMP      MAIN_LOOP
```

```
    LCALL      INT_INITIAL
    BSR        INTCON,GIE
    SLEEP
    LJUMP      MAIN
```

```
;
```

```
;系统初始化
```

```
;
```

```
INITIAL:
```

```
    BANKSEL    OSCCON
    LDWI        OSCCON_DEF
    STR         OSCCON
```

```
    BANKSEL    INTCON
    LDWI        INTCON_DEF
    STR         INTCON
```

```
    BANKSEL    PORTA
    LDWI        0X3F
    STR         PORTA
```

```

    BANKSEL    TRISA
    LDWI       TRISA_DEF
    STR        TRISA

    BANKSEL    WPUA
    LDWI       WPUA_DEF
    STR        WPUA

    BANKSEL    PSRCA
    LDWI       PSRCA_DEF
    STR        PSRCA

    BANKSEL    PSINKA
    LDWI       PSINKA_DEF
    STR        PSINKA

    BANKSEL    OPTION
    LDWI       OPTION_DEF
    STR        OPTION
;*****Clear SRAM*****
    BCR        STATUS,PAGE
    LDWI       0X40
    STR        FSR
CLEAR_RAM_BANK0_LOOP:
    CLRR       INDF
    INCR       FSR,F
    LDWI       80H
    XORWR      FSR,W
    BTSS       STATUS,Z
    LJUMP      CLEAR_RAM_BANK0_LOOP
    RET

;=====
;INT_INITIAL
;=====
INT_INITIAL:
    BANKSEL    TRISA
    BSR        TRISA,2           ;端口设置为输入
    BSR        OPTION,INTEDG     ;中断方式设置为上升沿
    BCR        INTCON,INTF       ;中断标志清零
    BSR        INTCON,INTE       ;中断使能
    RET

;=====
;DELAY_10MS 16MHZ/4T
;=====

```


DELAY_10MS:

LDWI H'C8'

STR TEMP

DELAY_10MS_LOOP1:

LDWI H'32'

STR TEMP1

DELAY_10MS_LOOP2:

CLRWDT

DECRSZ TEMP1,F

LJUMP DELAY_10MS_LOOP2

DECRSZ TEMP,F

LJUMP DELAY_10MS_LOOP1

RET

;

END

; 汇编程序结束

Fremont Micro Devices (SZ) Limited

#5-8, 10/F, Changhong Building, Ke-Ji Nan 12 Road, Nanshan District, Shenzhen, Guangdong 518057

Tel: (86 755) 86117811

Fax: (86 755) 86117810

Fremont Micro Devices (Hong Kong) Limited

#16, 16/F, Blk B, Veristrong Industrial Centre, 34-36 Au Pui Wan Street, Fotan, Shatin, Hong Kong

Tel: (852) 27811186

Fax: (852) 27811144

Fremont Micro Devices (USA), Inc.

42982 Osgood Road Fremont, CA 94539

Tel: (1-510) 668-1321

Fax: (1-510) 226-9918

Web Site: <http://www.fremontmicro.com/>

* Information furnished is believed to be accurate and reliable. However, Fremont Micro Devices, Incorporated (BVI) assumes no responsibility for the consequences of use of such information or for any infringement of patents of other rights of third parties which may result from its use. No license is granted by implication or otherwise under any patent rights of Fremont Micro Devices, Incorporated (BVI). Specifications mentioned in this publication are subject to change without notice. This publication supersedes and replaces all information previously supplied. Fremont Micro Devices, Incorporated (BVI) products are not authorized for use as critical components in life support devices or systems without express written approval of Fremont Micro Devices, Incorporated (BVI). The FMD logo is a registered trademark of Fremont Micro Devices, Incorporated (BVI). All other names are the property of their respective own.