FT62F21X

Application note

目录

FT60F01x Timer0 应用

1 Timer0 相关寄存器的设置

定时器 0 为 8 位，可配置为计数器或定时器使用，当作为外部事件（T0CKI）计数器时，可以配置为上升沿或者下降沿计数。作为定时器时，其计数时钟为系统时钟的 4 分频，即每一指令周期递增一次。有一个与 WDT 共用的 8 位预分频器，PSA 为 0 时该预分频器分配给定时器 0 使用。

相关寄存器的各个位定义如下：

1）TMR0 寄存器

Bit	7	6	5	4	3	2	1	0
Name	TMR0[7:0]							
Reset	xxxx xxxx							

Bit7~Bit0 Timer 0 计数结果寄存器

2.）OPTION 寄存器

Bit	7	6	5	4	3	2	1	0
Name	/PAPU	INTEDG	T0CS	T0SE	PSA	PS2	PS1	PS0
Reset	1	1	1	1	1	1	1	1

Bit7：PORTA 口上拉使能位

1：上拉功能被禁止

0：上拉功能使能

Bit6：触发中断边沿选择位

1：PA2/INT 上升沿触发中断

0：PA2/INT 下降沿触发中断

Bit5：Timer0 时钟选择位

1：PA2/T0CKI管脚输入时钟

0：内部指令周期Fosc/4

Bit4: Timer0 时钟边沿选择位

1：PA2/T0CKI管脚由高到底变化时计数增加

0：PA2/T0CKI管脚由低到高变化时计数增加

Bit3：预分频分配位

1：预分频器分配给WDT

0：预分频器分配给Timer0

Bit2~Bit0 预分频大小设置位

Bit2：Bit0	Timer0 Rate	WDT Rate
000	1 : 2	1 : 1
001	1 : 4	1 : 2
010	1 : 8	1 : 4
011	1 : 16	1 : 8
100	1 : 32	1 : 16
101	1 : 64	1 : 32
110	1 : 128	1 : 64
111	1 : 256	1 : 128

3）INTCON 寄存器

Bit	7	6	5	4	3	2	1	0
Name	GIE	PEIE	T0IE	INTE	PAIE	T0IF	INTF	PAIF
Reset	0	0	0	0	0	0	0	0
Type	RW	RW	RW	RW	RW	RW	RW	RW

Bit7：全局中断使能

1：使能所有未屏蔽中断

0：禁止所有中断

Bit6：外设中断使能

1：使能所有未屏蔽中断

0：禁止所有外设中断

Bit5：定时器 0 溢出中断使能

1：使能定时器 0 中断

0：禁止定时器 0 中断

Bit4：外部中断使能

1：使能 PA2/INT 管脚外部中断

0：禁止 PA2/INT 管脚外部中断

Bit3：PORTA 端口变化中断

1：使能 PORTA 端口变化中断

0：禁止 PORTA 端口变化中断

Bit2：定时器 0 溢出中断标志位

1：Timer0 寄存器溢出（必须软件清零）

0：Timer0 寄存器未溢出

Bit1：PA2/INT 管脚外部中断标志位

1：PA2/INT 管脚外部中断已发生（必须软件清零）

0：PA2/INT 管脚外部中断未发生

Bit0：PORTA 端口变化中断标志位

1：PORTA<5:0>至少有一个端口状态发生了改变（必须软件清零）

0：PORTA<5:0>没有一个端口发生状态改变

4) T0CON 寄存器

Bit	7	6	5	4	3	2	1	0
Name	-	-	-	-	T0ON	T0CKRUN	T0CKSRC	
Reset	-	-	-	-	0	0	0	0
Type	RO-0	RO-0	RO-0	RO-0	RW	RW	RW	RW

Bit7~Bit4： 保留为，读 0

Bit3： 定时器 0 使能位

1 = 使能（default 值为 1，保持向前兼容）

0 = 禁止

Bit2： 当 T0 时钟不是选择指令时钟时，睡眠状态 T0CK 的运行控制位

1 = T0CK 睡眠时保持工作

0 = T0CK 睡眠时停止工作

Bit1~Bit0： T0 时钟源选择

00 = 指令时钟

01 = HIRC

2 定时时间长度设置

例如在系统时钟和 4T 模式下，定时时长计算公式如下：

$$定时时长=\frac{1}{系统时钟频率}*4*预分频值*255$$

3 应用范例

```
//*************************************************************
/*   文件名：Test_62F21X_Timer0.c
*    功能：    FT62F21X_Timer0 功能演示
*    IC:       FT62F21X SOP8
*    晶振：    16M/4T
*    说明：    DemoPortOut 输出 30Hz 占空比 50%的波形-Timer0 实现
*
* Memory: Flash 1KX14b, EEPROM 128X8b, SRAM 64X8b
*                                  FT62F21X SOP8
*                                  ----------------
*   DemoPortOut ---------- |1(PA4)                (PA3)8 |----------NC
*   NC---------------------|2(TKCAP)              (PA0)7 |----------NC
*   NC---------------------|3(VDD)                (PA1)6 |----------NC
*   NC---------------------|4(VSS)                (PA2)5 |----------DemoPortIn
*                                  ----------------
*/
//=============================================================
#INCLUDE   <FT62F21X.INC>;
;=============================================================
;RAM DEFINE
TEMP                EQU     0X40
TEMP1               EQU     0X41
TEMP2               EQU     0X42
W_TMP           EQU     0X4C
S_TMP               EQU     0X4D
;=============================================================================
;CONSTANT DEFINE
;=============================================================================
INTCON_DEF      EQU     B'00000000'     ;GIE, TMR0IE,
OPTION_DEF      EQU     B'00000000'     ;PORTA pull-ups are enable;Timer0 1:2
OSCCON_DEF      EQU     B'01110000'     ;16MHz INTERNAL OSC
WPUA_DEF        EQU     B'00000000'     ;
TRISA_DEF       EQU     B'00000000'     ;PA4-OUT

LSB             EQU     0
MSB             EQU     7
```

```
;=====================================================================
;USER DEFINE
;=====================================================================
#define DemoPortOut    PORTA,4
#define DemoPortIn     PORTA,2
;=====================================================================
;PROGRAM START
;=====================================================================
    ORG             0000H
    LJUMP           RESTART
    ORG             0004H
    STR             W_TMP
    SWAPR           STATUS,W
    STR             S_TMP
    BCR             STATUS,RP0

INT_RET:
    BANKSEL         INTCON
    BCR             INTCON,T0IF
    BANKSEL         PORTA
    LDWI        10H
    XORWR       PORTA,F             //取反 PA4


    SWAPR       S_TMP,0
    STR         STATUS
    SWAPR       W_TMP,1
    SWAPR       W_TMP,0
    RETI

;=====================================================================
;SYSTEM START
;=====================================================================
RESTART:
    BANKSEL     PORTA
    LCALL       INITIAL
    LCALL       TIMER0_INIT
    BANKSEL         INTCON
    BSR         INTCON,GIE
    BSR         INTCON,T0IE
MAIN_LOOP:
    NOP
    NOP
    LJUMP       MAIN_LOOP
```

```
;=====================================================================
;SYSTEM INITIAL
;=====================================================================
INITIAL:
    BANKSEL     PORTA
    LDWI        0X00
    STR         PORTA           ;Clear PortA
    BANKSEL     TRISA
    LDWI        TRISA_DEF       ;PA4-OUT
    STR         TRISA           ;SET IO Direction
    LDWI        WPUA_DEF
    STR         WPUA
    LDWI        OPTION_DEF
    STR         OPTION_REG      ;SET OPTION
    LDWI        OSCCON_DEF
    STR         OSCCON          ;SET OSCCON
    BANKSEL     PORTA
    LDWI        INTCON_DEF
    STR         INTCON
;   BSR         MSCKCON,SLVREN
CLEAR_RAM:
    LDWI        40H
    STR         FSR
CLEAR_RAM_LOOP:
    CLRR        INDF
    INCR        FSR,F
    LDWI        80H
    XORWR       FSR,W
    BTSS        STATUS,Z
    LJUMP       CLEAR_RAM_LOOP
    RET
;=====================================================================
;Timer0 init
;设置 TMR0 定时时长 16.384ms=(1/16000000)*4*256*255(16M-2T-PSA 1:256- TMR0=255 溢出)
;TMR0 = 0;
;=====================================================================
TIMER0_INIT:
    BANKSEL     OPTION
    LDWI        B'00000111'
    STR         OPTION
    BCR         INTCON,T0IF

    RET
;=====================================================================
```

```
;DELAY_10MS 16MHZ/4T
;=====================================================================
DELAY_10MS:
    LDWI        H'28'
    STR         TEMP1
    LDWI            H'0F'
    STR         TEMP2
DELAY_10MSLOOP3:
    CLRWDT
    DECRSZ       TEMP2,F
    LJUMP        DELAY_10MSLOOP3
    DECRSZ       TEMP1,F
    LJUMP        DELAY_10MSLOOP3
    RET

 END
```

Fremont Micro Devices (SZ) Limited

#5-8, 10/F, Changhong Building, Ke-Ji Nan 12 Road, Nanshan District, Shenzhen, Guangdong 518057
Tel: (86 755) 86117811
Fax: (86 755) 86117810

Fremont Micro Devices (Hong Kong) Limited

#16, 16/F, Blk B, Veristrong Industrial Centre, 34-36 Au Pui Wan Street, Fotan, Shatin, Hong Kong
Tel: (852) 27811186
Fax: (852) 27811144

Fremont Micro Devices (USA), Inc.

42982 Osgood Road Fremont, CA 94539
Tel: (1-510) 668-1321
Fax: (1-510) 226-9918

Web Site: http://www.fremontmicro.com/