

FT62F21X

Application note

FMD Confidential

目录

1	IR 介绍	3
2	应用范例	3

FT62F21X IR Receive 应用

1 IR 介绍

一个通用的红外遥控系统由发射和接收两大部分组成，如图 1 所示：




图 1

发射部分主要包括键盘矩阵、编码调制、红外发射管；接收部分包括光、电信号的转换以及放大、解调、解码电路。

举例来说，通常我们家电遥控器信号的发射，就是将相应按键所对应的控制指令和系统码（由 0 和 1 组成的序列），调制在 32~56kHz 范围内的载波上（目的为：抗干扰及低功率），然后经放大（接三极管）、驱动红外发射管（透明的头）将信号发射出去。

本讲解以 IC FT60F011A SOP8 为示范，采用一体的红外接收头，接收头输出脚连到 MCU 的 IO 口，IO 口通过识别高低电平时间长短来解码，当收到的数据是合法的，指示 LED 的状态（开与关）会翻转一次。接收的 IO 口使用电平变化中断来识别信号，并使用定时器记录电平的时间长短。

本程序 IR 接收与 LED 所对应的 IO 引脚：

```
#define IRRIO      RA2
```

```
#define LED        RA4
```

2 应用范例

```

//=====
; * 文件名:  ASM_FT62F21X_IR_Receive.ASM
; * 功能:    FT62F21X 红外接收 功能演示
; * IC:     FT62F211    SOP8
; * 内部:    16M/4T
; * 说明:    接受红外发送过来的两对互补字节，并与自己所需要的比对，确认正确后
; *          执行所需操作
; *
; *
; *          FT62F211  SOP8
; *          -----
; * ir_rx-----|1(PA4)   (PA3)10|-----NC
; * NC-----|2(TKCAP) (PA0)9|-----NC
  
```

```

;* VDD-----|3(VDD)   (PA1)8|-----NC
;* GND-----|4(VSS)   (PA2)7|-----NC
;*
//=====
#include <FT62F21X.INC>;
//=====
;RAM DEFINE
//=====
TEMP EQU 0X40
TEMP1 EQU 0X41
TEMP2 EQU 0X42
IRDATTEMP EQU 0X43
READPIN EQU 0X44
buff EQU 0X45
#define f_finish buff,0
#define f_receive buff,1
ir_counttemp EQU 0X46
ir_count EQU 0x47
countbyte EQU 0x48
IRDATE0 EQU 0X49
IRDATE1 EQU 0X4A
IRDATE2 EQU 0X4B
IRDATE3 EQU 0X4C
W_TMP EQU 0X70
S_TMP EQU 0X71
//=====
;CONSTANT DEFINE
//=====
INTCON_DEF EQU B'00000000' ;禁止所有中断

OSCCON_DEF EQU B'01110000' ;16MHz

WPUA_DEF EQU B'00010000' ;弱上拉的开关, 0-关, 1-开

TRISA_DEF EQU B'00010000' ;输入输出设置, 0-输出, 1-输入

PSRCA_DEF EQU B'00001111' ;源电流设置最大

PSINKA_DEF EQU B'00000011' ;灌电流设置最大

OPTION_DEF EQU B'00001000' ;Bit3=1 WDT MODE,PS=000=1:1 WDT RATE
;Bit7(PAPU)=0 由 WPUA 决定是否上拉
//=====
;USER DEFINE

```

```

=====
;
#define  ir_rx      PORTA,4
;
=====
;PROGRAM START
;
=====
    ORG      0x0000      ; 单片机复位向量入口
    LJUMP    RESTART    ; 跳转到主程序入口
    ORG      0x0004      ; 中断复位向量入口
    LJUMP    INT_PROGRAM
;
=====
;中断处理程序
;
=====
INT_PROGRAM:
    STR      W_TMP      ; 保存 W 寄存器
    SWAPR    STATUS,W    ; 保存 STATUS 寄存器
    STR      S_TMP

    BANKSEL  INTCON
    BTSC     INTCON,T0IF
    LJUMP    TM0Interrupt
    LJUMP    INT_RET

TM0Interrupt:
    BCR      INTCON,T0IF
    INCR     ir_count,1
    LJUMP    INT_RET

INT_RET:
    SWAPR    S_TMP,0
    STR      STATUS      ; 恢复 STATUS 寄存器
    SWAPR    W_TMP,1
    SWAPR    W_TMP,0      ; 恢复 W 寄存器
    RETL     ; 中断返回
;
=====
;SYSTEM START
;
=====
RESTART:
    BANKSEL  PORTA
    LCALL    INITIAL
    LCALL    TIMER0_INITIAL
;
=====
;主程序
;
=====
MAIN:

```

```

CLRWDT
LCALL    scanky_ir
LCALL    ir_data_compare
LJUMP    MAIN
;=====
;系统初始化
;=====
INITIAL:
    BANKSEL    OSCCON
    LDWI        OSCCON_DEF
    STR         OSCCON

    BANKSEL    INTCON
    LDWI        INTCON_DEF
    STR         INTCON

    BANKSEL    PORTA
    LDWI        0X00
    STR         PORTA

    BANKSEL    TRISA
    LDWI        TRISA_DEF
    STR         TRISA

    BANKSEL    WPUA
    LDWI        WPUA_DEF
    STR         WPUA

    BANKSEL    PSRCA
    LDWI        PSRCA_DEF
    STR         PSRCA

    BANKSEL    PSINKA
    LDWI        PSINKA_DEF
    STR         PSINKA

    BANKSEL    OPTION
    LDWI        OPTION_DEF
    STR         OPTION

;*****Clear  SRAM*****
    BCR         STATUS,PAGE
    LDWI        0X40
    STR         FSR
CLEAR_RAM_BANK0_LOOP:

```

```

CLRR      INDF
INCR      FSR,F
LDWI      80H
XORWR     FSR,W
BTSS      STATUS,Z
LJUMP     CLEAR_RAM_BANK0_LOOP
RET

```

```

;=====
;PA3_Level_Change_INITIAL
;=====

```

```

PA3_Level_Change_INITIAL:

```

```

    BANKSEL TRISA
    BSR      TRISA,3           ;端口设置为输入

```

```

    BANKSEL PORTA
    LDR      PORTA,W
    BANKSEL READPIN
    STR      READPIN

```

```

    BANKSEL INTCON
    BCR      INTCON,PAIF      ;中断标志清零
    BANKSEL IOCA
    BSR      IOCA,IOCA3
    BANKSEL INTCON
    BSR      INTCON,PAIE     ;中断使能
    RET

```

```

;=====
;TIMER0_INITIAL
;设置 TMR0 定时时长 256Us=(1/16000000)*4*4*255
;=====

```

```

TIMER0_INITIAL:

```

```

    BANKSEL OPTION
    LDWI      0X01
    STR      OPTION
    BANKSEL TMR0
    LDWI      0
    STR      TMR0
    BANKSEL INTCON
    BSR      INTCON,T0IE
    BSR      INTCON,PEIE
    BSR      INTCON,GIE
    RET

```

```
=====
;scanky_ir
```

```
;采集红外接收器收到的数据
=====
```

```
scanky_ir:
```

```
    BTSC    ir_rx
```

```
    RET
```

```
    CLRR    ir_count
```

```
    BSR     f_finish
```

```
scanky_ir_one:
```

```
    CLRWDT
```

```
    LDWI    0x27
```

```
    SUBWR   ir_count,0
```

```
    BTSC    STATUS,0
```

```
    LJUMP   scanky_ir_wrong
```

```
    BTSS    ir_rx
```

```
    LJUMP   scanky_ir_one
```

```
    LDR     ir_count,0
```

```
    STR     ir_counttemp
```

```
;判断启动时候的 9ms 的低电平
```

```
    CLRR    ir_count
```

```
    LDWI    0x20
```

```
    SUBWR   ir_counttemp,0
```

```
    BTSS    STATUS,0
```

```
    LJUMP   scanky_ir_wrong
```

```
    LDWI    0x27
```

```
    SUBWR   ir_counttemp,0
```

```
    BTSC    STATUS,0
```

```
    LJUMP   scanky_ir_wrong
```

```
scanky_ir_two:
```

```
    CLRWDT
```

```
    LDWI    0x13
```

```
    SUBWR   ir_count,0
```

```
    BTSC    STATUS,0
```

```
    LJUMP   scanky_ir_wrong
```

```
    BTSC    ir_rx
```

```
    LJUMP   scanky_ir_two
```

```
    LDR     ir_count,0
```

```
    STR     ir_counttemp
```

```
;判断启动时候的 4.5ms 的高电平
```

```
    CLRR    ir_count
```

```
    LDWI    0x0f
```

```
    SUBWR   ir_counttemp,0
```

```
    BTSS    STATUS,0
```



```
LJUMP  scanky_ir_wrong
LDWI   0x13
SUBWR  ir_counttemp,0
BTSC   STATUS,0
LJUMP  scanky_ir_wrong
```

```
CLRR   IRDATE0
CLRR   IRDATE1
CLRR   IRDATE2
CLRR   IRDATE3
BCR    f_receive
LDWI   0x21
STR    countbyte
```

scanky_ir_thr:

```
CLRWDT
LDWI   0x05
SUBWR  ir_count,0
BTSC   STATUS,0
LJUMP  scanky_ir_wrong
BTSS   ir_rx
LJUMP  scanky_ir_thr
```

```
LDR    ir_count,0
STR    ir_counttemp
CLRR   ir_count
LDWI   0x05
SUBWR  ir_counttemp,0
BTSC   STATUS,0
LJUMP  scanky_ir_wrong
```

scanky_ir_fou:

```
CLRWDT
LDWI   0x08
SUBWR  ir_count,0
BTSC   STATUS,0
LJUMP  scanky_ir_wrong
BTSS   ir_rx
LJUMP  scanky_ir_fou
```

```
DECRSZ countbyte,1
LJUMP  $+2
LJUMP  scanky_ir_receive_end
```

```
BCR    STATUS,0
```

```
RRR    IRDATE3,1
RRR    IRDATE2,1
RRR    IRDATE1,1
RRR    IRDATE0,1
LDR    ir_count,0
STR    ir_counttemp
CLRR   ir_count
LDWI   0x05
SUBWR  ir_counttemp,0
BTSS   STATUS,0
LJUMP  scanky_ir_thr
BSR    IRDATE3,7
LJUMP  scanky_ir_thr
```

scanky_ir_wrong:

```
BCR    f_receive
BCR    f_finish
RET
```

scanky_ir_receive_end:

```
BSR    f_receive
BCR    f_finish
RET
```

```
=====
;
;ir_data_compare
; 比较采集到的数据
;
=====
```

ir_data_compare:

```
BTSS   f_receive
RET
BCR    f_receive

LDWI   0x55
XORWR  IRDATE0,0
BTSS   STATUS,2
RET
```

```
LDWI   0xaa
XORWR  IRDATE1,0
BTSS   STATUS,2
RET
```

```
LDWI   0x01
XORWR  IRDATE2,0
```

BTSS STATUS,2

RET

LDWI 0xfe

XORWR IRDATE3,0

BTSS STATUS,2

RET

NOP ;如果是所想要的数值则。。。

RET

=====

END

; 汇编程序结束

Fremont Micro Devices (SZ) Limited

#5-8, 10/F, Changhong Building, Ke-Ji Nan 12 Road, Nanshan District, Shenzhen, Guangdong 518057

Tel: (86 755) 86117811

Fax: (86 755) 86117810

Fremont Micro Devices (Hong Kong) Limited

#16, 16/F, Blk B, Veristrong Industrial Centre, 34-36 Au Pui Wan Street, Fotan, Shatin, Hong Kong

Tel: (852) 27811186

Fax: (852) 27811144

Fremont Micro Devices (USA), Inc.

42982 Osgood Road Fremont, CA 94539

Tel: (1-510) 668-1321

Fax: (1-510) 226-9918

Web Site: <http://www.fremontmicro.com/>

* Information furnished is believed to be accurate and reliable. However, Fremont Micro Devices, Incorporated (BVI) assumes no responsibility for the consequences of use of such information or for any infringement of patents of other rights of third parties which may result from its use. No license is granted by implication or otherwise under any patent rights of Fremont Micro Devices, Incorporated (BVI). Specifications mentioned in this publication are subject to change without notice. This publication supersedes and replaces all information previously supplied. Fremont Micro Devices, Incorporated (BVI) products are not authorized for use as critical components in life support devices or systems without express written approval of Fremont Micro Devices, Incorporated (BVI). The FMD logo is a registered trademark of Fremont Micro Devices, Incorporated (BVI). All other names are the property of their respective own.